
5 EXAMPLE Nº 5: EARTH DAM 

5.1 AIM 

 

The aim of this example is to show the combined possibilities of the utilities that 
CivilFEM provides for:  

 Seepage analysis, including the calculation of the saturation line. 

 Slope stability analysis. 

At the same time as it shows its application to successfully complete study of an 
earth dam, making it possible to the engineer to design the slopes of a dam of these 
characteristics.  

The example shown in this text has a certain complication and, as it will be seen, it 
requires a lengthy study. 

 

5.2 DESCRIPTION OF THE EXAMPLE 

 

In the following figure the geometry of the dam and the materials characteristics are 
shown. 

 

 

 

The geometric data of the model are:  

 

Saturation
curve

Water surface

   Top of dam:

   Ytop ttop

tdown

tup

Impermeable boundary

k, , c, , E,   Xini

Yfondation  =  120.00 m  

Ywater  =  190.00 m  

Ytop  =  192.00 m  

Yexit   =  Yfoundation   + 20.00 m  

tup  =  2.00 m  

tdown  =  2.10 m  

ttop  =  10.00 m  

xini  =  20.00 m  

 

The properties of land RANDOM FILL are:  

 

k  =  10-7  m/s  

d  =  30  º  

c  =  2  T/m2  

d  =  2.0  T/m3  

E  =  2·104  T/m2  

 =  0.30   

 

5.3 RESULTS TO BE OBTAINED 

 

1. Saturation line and flow through the dam.  

2. Pore pressures distribution.  

3. Calculate the safety factor of the downstream slope:  

3.1. According to Bishop.  

3.2. According to Janbu.  

4. Analyse the dam using finite elements method. The hydrostatic pressure will 
be considered as a pressure on the upstream slope.  

5. Calculate the law of vertical effective pressures.  

6. Obtain the safety factor, according to the circles used in section 3, using the 
effective pressures obtained in the finite elements method. 

 

5.4 CALCULATION LOG 

5.4.1 Introduction 

 

At the beginning of the resolution of section 1 only the geotechnical materials with 
low permeability are defined, since the aim of this example is to perform a seepage 
analysis. The values of these permeabilities are modified (KX, KY, KZ).  

The parameters that define the geometry of the model are set. Next, the model is 

created, without boolean operations and with a structure of down to top (Keypoints  

Lines  Areas  Elements and nodes). 

The saturation line is then defined. It is important to establish the type of boundary 
condition of the starting and ending points of this line. In this case, the starting point 
stays fixed at the same location (FIXED) as the water level, and the ending point 
slides along the downstream slope with a seepage condition (EXIT). 

Once the saturation line is set, the finite elements model is generated from a group of 
lines, meshing the areas these lines define with PLANE42 elements. 

A PLANE42 element type (structural 2D solids) is automatically created when a 2-D 
seepage analysis is actived. The degree of freedom for this calculation is the 
hydraulic potential H, that CivilFEM will internally associate with the displacement in 
the X direction (UX) for PLANE42 elements. 

The seepage boundary conditions are then set. These conditions are the hydraulic 
head upstream and a seepage condition downstream.  

The model is finally solved and the results of the flow are obtained.   

In section 2 the pore pressures contour lines are calculated. This simply consists in 
obtaining a new result from the previous calculations.  

It is in this step when the results of pore pressures are transferred to the slope 
stability utility that will be later used. For this, a file is generated (jobname.PRESS) 
that contains the pore pressures results at each node of the model.  

Because of the similarity between the different sections of this example, and in order 
to simplify the code, in the following sections an APDL macro is created. At the 
beginning of this macro the structural materials are defined, changing their properties 
to those of the specifications of the example. It is not necessary to modify the 
permeability parameters of these materials, because they are only going to be used 
in structural analysis and slope stability calculations. A structural quadratic 2D 
element (PLANE82) with plane strain is used for the calculations. Geometrical 
parameters are also defined, and from these the keypoints are created. Areas are 
generated from the keypoints and finally these areas are meshed. 

In section 3 a slope stability analysis is requested by means of two classic methods: 
Bishop and Janbu. 

The classical methods study the stability of a given sliding surface using the Mohr-
Coulomb theory and the static equilibrium of moments or forces (or both). The 
methods compare the destabilizing actions and the stabilizing ones, and they obtain 
a ratio which is the safety factor.   

Once the model is created, the pressures acting on the surface are applied. Later the 
model is captured to perform the slope stability analysis; to do so 2 points that define 

the outer surface (the part of the model that may slide) and an intermediate point to 
show which part of the boundary is the one selected. 

Bishop:  

The grid of centres of the circles that will be used as sliding surfaces are entered by 
coordinates or by keypoints, and also the number of divisions at each side. Next, the 
points that define the mesh of tangents that will define the radii of these circles are 
also set. 

The calculation of the safety factor is an iterative. The parameters that control it 
(tolerance, maximum number of iterations) can be changed if needed, but in this 
example the default values will be used. 

The pore pressure values used are the ones obtained from the seepage analysis. 
These values are read from the nodal data included in the jobname.PRESS file. 

Janbu:  

The sliding surfaces definition in this method is different from the definition for 
Bishop’s method. In this case it is necessary to define the points that will set the 
polygonal sliding surfaces. 

The polygonal line has to be convex and must begin and end outside the model 
(CivilFEM will discard invalid polygonals). 

In section 4 The actions on the model are self weight and the water pressure on the 
slope, when this one is submerged. Water pressures must be entered as a an 
external pressure. 

For section 5 CivilFEM’s utility to import pore pressure results from a seepage 
analysis will be used. The meshes of the seepage model and the stability analysis or 
finite element analysis model do not need to be the same. 

The effective pressures are calculated in the structural analysis subtracting the pore 
pressures.  

In section 6, in order to compare the results given by the classical methods with the 
ones provided by the finite element analysis, CivilFEM calculates the equivalent 
safety factor at each point of the sliding surface.  

The normal and tangential stresses are obtained from the results of the finite element 
analysis. The pore pressures distribution is defined in the same way as for the 
classical methods.  

5.4.2 Section 1: Line of saturation and filtered volume (Generation of 
the model)  

 
FINISH 

~CFCLEAR,,1 

 

~CFACTIV,GETC,Y ! Activate geotechnical module 

~SEEPAGE,2D 

 

/TITLE, Slope stability analysis 

 

! CivilFEM Setup 

~UNITS,SI 

 

/PREP7 

! CivilFEM Preprocessor 

! ----------------------------------------------------------------------- 

! Materials   

~CFMP,1,LIB,SOIL,,SM           ! Layer 1   Sand Silty 

~CFMP,1,SOIL,KX,,1e-007 

~CFMP,1,SOIL,KY,,1e-007 

~CFMP,1,SOIL,KZ,,1e-007 

 

~CFMP,2,LIB,ROCK,,Molena,      ! Layer 2   Molena (Sedimentary Rock) 

~CFMP,2,ROCK,KX,,1e-012 

~CFMP,2,ROCK,KY,,1e-012 

~CFMP,2,ROCK,KZ,,1e-012 

 

! Model definition 

! Parameters 

Yfoundation = 120.00 

Ywater      = 190.00 

Ytop        = 192.00 

Yexit       = Yfoundation + 20.00 

tup         =   2.00 

tdown       =   2.10 

ttop        =  10.00 

Xini        =  20.00 

Xleft       = (Ywater - Yfoundation)*tup + Xini 

Xleft1      = (Ytop - Yfoundation)*tup + Xini 

Xright      = Xleft1 + ttop + (Ytop - Yfoundation)*tdown 

Xexit       = Xleft1 + ttop + (Ytop - Yexit)*tdown 

XEndPnt     = 200.00 

XbaseR      = 500.00 

 

X1  = 0          $ Y1  = 0 

X2  = XbaseR     $ Y2  = 0 

X3  = 0          $ Y3  = Yfoundation 

X4  = Xini       $ Y4  = Y3 

X5  = Xright     $ Y5  = Y3 

X6  = X2         $ Y6  = Y3 

X7  = Xleft      $ Y7  = Ywater 

X8  = Xleft1     $ Y8  = Ytop 

X9  = X8 + ttop  $ Y9  = Y8 

X10 = Xexit      $ Y10 = Yexit 

 

! KP's definition 

*DO,II,1,10 

  K,II,X%II%,Y%II% 

*ENDDO 

 

! Lines 

L,1,2 

L,2,6 

L,1,3 

L,3,4 

L,4,5 

L,5,6 

L,4,8 

L,8,9 

L,9,5 

 

! Plot lines with their Numbers 

/PNUM,KP,1 

/PNUM,LINE,1 

LPLOT 

 

! Defines the saturation line 

X1      = X7      ! X coordinate of the initial point of the saturation 

                  ! line (KP,7) 

Y1      = Y7      ! Y initial point 

Z1      = 0       ! Z initial point 

X2      = X10     ! X coordinate of the end point of the saturation line 

Y2      = Y10     ! Y end point 

Z2      = 0       ! Z end point 

NPOINTS = 5       ! Number of points to be generated along the saturation 

                  ! line 

Line1   = 7       ! The line number on which the initial point of the 

                  ! saturation line leans 

Line2   = 9       ! The line number on which the end point of the 

                  ! saturation line leans 

MAT     = 1       ! Material used to mesh the area from which the 

                  ! saturation line belongs to 

 

! SATURATION LINE 

~WATTAB,X1,Y1,Z1,X2,Y2,Z2,MAT,NPOINTS,Line1,Line2,FIXED,EXIT 

/REPLOT 

 

 
 

! Generate the finite element model from a line set 

MAT   = 2     ! Material to be used 

ESIZE = 7     ! Element size 

L1    = 3   $  L2    = 4  $   L3    = 5    $   L4   = 6 

L5    = 2   $  L6    = 1 

~SEEPMOD,ADD,MAT ,,ESIZE,L1,L2,L3,L4,L5,L6       ! Area 1 

 

k-points and lines definition

L11 2

L2

2

6

L3

1

3 L43 4 L54 5 L65 6

L7

4

11L88 9

L9
5

12

11

13

14L1214

16L1416

L15

17
12

MAT   = 1 

ESIZE = 5 

L1    = 7   $  L2    = 5   $   L3   = 9 

~SEEPMOD,ADD,MAT,WT,ESIZE,L1,L2,L3               ! Area 3 

 

FINISH 

 

 

 

5.4.3 Section 1: Solution  

 
/SOLU  

! Generates a constant piezometric head as a boundary condition 

 HEAD1 = Ywater 

~DLHEAD,ADD,7,HEAD1 

~DLSEEP,ADD,9 

 

! Water Table solver 

~WTSOLVE 

 

 

 

Saturation line definition

5.4.4 Section 1: Postprocess 

 
/POST1 

! Calculates the flux through the saturation line 

Node1=NODE(194.444,0,0) 

Node2=NODE(138.991,159.490,0) 

Node3=NODE(131.8176,12.7631,0) 

~PLSEEP,Q,GRAPH,N,Node1,Node2,Node3   ! Total flux 

~PLSEEP,Q,VECTOR                ! Vectorial flux 

 

5.4.5 Section 2: Pore pressures distribution  
 

/GRAPHICS,FULL    ! Display all model geometry and results 

~ISOBAR    ! Plot pore water pressure results 

 

! Transfers the pore water pressure results to the slope calculation 

! routines  

~WTSLP     ! This command automatically generates a text file called 

           ! jobname.PRESS. 

 

5.4.6 Section 3.1: Calculation by Bishop’s method (model generation)  

 
FINISH 

~CFCLEAR,,1 

~SEEPAGE,NONE 

 

~CFACTIV,GETC,Y  ! Activate geotechnical module 

/TITLE,Slope stability analysis 

 

! CivilFEM Setup 

~UNITS,SI 

 

/PREP7   

! CivilFEM Preprocessor 

! ---------------------------------------------------------------------- 

MACRO_PREP7_FOR_SLOPE_MOD    ! macro that includes: Materials, Element type, 

                             ! model & mesh definition 

 

! Boundary Conditions 

gammaw = 9810  ! N/m^3 

SFL,5,PRES,gammaw*(Y7-Y3),   ! Vertical load over the soil 

SFGRAD,PRES,0,Y,Y3,-gammaw   ! Pressure over the slope 

SFL,11,PRES,gammaw*(Y7-Y3), 

SFTRANS                      ! Transfer the solid model surface loads to 

                             ! the FEM 

/PSF,PRES,NORM,2,0,1         ! Shows surface load symbols on model displays 

/PBF,DEFA, ,1                ! Shows magnitude of body force loads on 

                             ! displays 

/PIC,DEFA, ,1                ! Shows initial conditions on elements as 

                             ! contours on displays 

/PSYMB,DOT,1                 ! Larger symbols displayed for node and 

                             ! keypoint locations 

EPLOT 

 

! Define the model contour 

~SLPINK,3,6,5           ! Capture the model 

  

! Introduction of points defining the circle mesh 

~SLPCIR,182.0,210.0,282.0,162.0,235.0,302.0,10,10 

! Introduction of points defining the tangential mesh 

~SLPTAN,172.0,191.0,325.0,120.0,323.0,100.0,39.0,100.0,10 

 

~SLPOPT,,,,,,3    ! water pressure will be read from jobname.PRESS 

                  ! (~WTSLP) in the command ~SLPSOL 

 

 
 

5.4.7 Section 3.1: Results 

 
/POST1 

! Solving slope 

~SLPSOL,1,2, ! Bishop Calculation. Slides towards right 

 

 

 

5.4.8 Section 3.2: Calculation by Janbu’s method (model generation)  

 
FINISH 

~CFCLEAR,,1 

 

~CFACTIV,GETC,Y  ! Activate geotechnical module 

/TITLE,Slope estability analysis 

 

! CivilFEM Setup 

~UNITS,SI 

 

/PREP7   

! CivilFEM Preprocessor 

! ---------------------------------------------------------------------- 

MACRO_PREP7_FOR_SLOPE_MOD     ! macro that includes: Materials, Element 

                              ! type, model & mesh definition. 

 

! Boundary Conditions 

gammaw = 9810  ! N/m^3 

SFL,5,PRES,gammaw*(Y7-Y3),   ! Vertical load over the soil 

SFGRAD,PRES,0,Y,Y3,-gammaw   ! Pressure over the slope 

SFL,11,PRES,gammaw*(Y7-Y3), 

SFTRANS                      ! Transfer the solid model surface loads to 

                             ! the FEM 

/PSF,PRES,NORM,2,0,1         ! Shows surface load symbols on model displays 

/PBF,DEFA, ,1                ! Shows magnitude of body force loads on 

                             ! displays 

/PIC,DEFA, ,1                ! Shows initial conditions on elements as 

                             ! contours on displays 

/PSYMB,DOT,1                 ! Larger symbols displayed for node and 

                             ! keypoint locations 

EPLOT 

 

! Points that define the model contour 

~SLPINK,3,6,5           ! Capture the model 

 

! Defines the points that generate the Janbu’s polygonal 

~SLPPOL,100,230,200,170,140,130,120,110,80,50, 

~SLPPOL,250,230,200,170,140,130,120,110,80,50, 

~SLPPOL,470,230,200,170,140,130,120,110,80,50, 

 

~SLPOPT,,,,,,3    ! water pressure will be read from jobname.PRESS (~WTSLP) 

                  ! in the command ~SLPSOL 

 

5.4.9 Section 3.2: Results  

 
/POST1 

! Solving slope 

~SLPSOL,2,2   ! Janbu Calculation. Slides towards right 

 

5.4.10 Section 4: Calculation by FEM (model generation)  

 
FINISH 

~CFCLEAR,,1 

 

/TITLE,FEM analysis 

 

! CivilFEM Setup 

~UNITS,SI 

 

/PREP7 

! CivilFEM Preprocessor 

! ---------------------------------------------------------------------- 

! macro that includes: Materials, Element type, model & mesh definition 

MACRO_PREP7_FOR_SLOPE_MOD 

! Boundary conditions 

DL,1,,UY,0 

DL,2,,UX,0 

DL,6,,UX,0 

DTRAN                  ! Transfers solid model DOF constraints to the FEM 

EPLOT 

 

5.4.11 Section 4: Calculation  

 
/SOLU 

! Gravity 

ACEL,,9.8                    ! Self Weight 

 

gammaw = 9810  ! N/m^3 

SFL,5,PRES,gammaw*(Y7-Y3),   ! Vertical load over the soil 

SFGRAD,PRES,0,Y,Y3,-gammaw   ! Pressure over the slope 

SFL,11,PRES,gammaw*(Y7-Y3), 

SFTRANS                      ! Transfer the solid model surface loads to 

                             ! the FEM 

/PSF,PRES,NORM,2,0,1         ! Shows surface load symbols on model displays 

/PBF,DEFA, ,1                ! Shows magnitude of body force loads on 

                             ! displays 

/PIC,DEFA, ,1                ! Shows initial conditions on elements as 

                             ! contours on displays 

/PSYMB,DOT,1                 ! Larger symbols displayed for node and 

                             ! keypoint locations 

EPLOT 

 

 

! Solving the finite element model 

SOLVE  

 

 Pressure over the slope

Water pressure is considered
as a surface pressure on the
upstream slope

5.4.12 Section 4: Postprocess 

 
/POST1 

~CFSET,,1 

/GRAPHICS,FULL          ! Display all model geometry and results 

/DSCALE,1,1.0 

PLNSOL,S,Y 

 

5.4.13 Section 5: Effective pressures 

 
~EFFPRES,Y    ! Effective Pressures 

 

5.4.14 Section 6: Safety factor  

 
FINISH 

~CFCLEAR,,1 

 

~CFACTIV,GETC,Y ! Activate geotechnical module 

/TITLE,FEM analysis 

 

! CivilFEM Setup 

~UNITS,SI 

 

/PREP7 

! CivilFEM Preprocessor 

! ---------------------------------------------------------------------- 

! macro that includes: Materials, Element type, model & mesh definition 

MACRO_PREP7_FOR_SLOPE_MOD 

 

! Boundary conditions 

DL,1,,UY,0 

DL,2,,UX,0 

DL,6,,UX,0 

DTRAN                  ! Transfers solid model DOF constraints to the FEM 

EPLOT 

 

! Points that define the model contour 

~SLPINK,3,6,5           ! Capture the model 

  

! Introduction of points defining the circle mesh 

~SLPCIR,182.0,210.0,282.0,162.0,235.0,302.0,10,10 

 

! Introduction of points defining the tangential mesh 

~SLPTAN,172.0,191.0,325.0,120.0,323.0,100.0,39.0,100.0,10 

 

~SLPOPT,,,,,,3    ! water pressure will be read from jobname.PRESS 

                  ! (~WTSLP) in the command ~SLPSOL 

 

5.4.15 Section 6: Calculation  

 

The log file is the same as the one used in the section  5.4.11 

 



  EXAMPLE Nº 5: EARTH DAM 

5-13 Manual of Advanced Examples. Ingeciber, S.A.
© 

Ver. 2025 

5.4.16 Section 6: Postprocess 

 
/POST1 

~CFSET,,1 

!Solving slope 

~SLPSOL,10,2     ! FEM Calculation. Slides towards right 

 

5.4.17 Auxiliary macro  

 

This macro is used in some of the APDL files of the previous sections.  

 
! MACRO_PREP7_FOR_SLOPE_MOD: macro that includes: Materials, Element type, 

! model & mesh definition. 

! Used in slope stability problem 5 sections: BISHOP, JANBU, FEM 

 

! Materials 

~CFMP,1,LIB,SOIL,,SM       ! Layer 1   Sand Silty 

~CFMP,1,SOIL,EXST,,1.96E8  ! Young´s Modulus = 2.10E4 T/m^2 = 1.96E8 N/m^2 

~CFMP,1,SOIL,PHIMCEFF,,35  ! PHIMCeff = 35º 

~CFMP,1,SOIL,GAMD,,19600   ! GAMD = 2T/m^3 =2*9800N/m^3 

~CFMP,1,SOIL,CMCEFF,,19600 ! CMCEFF = 2T/m^2 =2*9800N/m^2 

~CFMP,1,SOIL,DELEFF,,35/2 

 

~CFMP,2,LIB,ROCK,,Molena   ! Layer 2   Molena (Sedimentary Rock) 

 

! Ansys Preprocessor 

ET,1,PLANE82               ! Element type 1: Solid-2D 

KEYOPT,1,3,2               ! Plane Strain 

 

! Model definition 

! Parameters 

Yfoundation = 120.00 

Ywater      = 190.00 

Ytop        = 192.00 

Yexit       = Yfoundation + 20.00 

tup         =   2.00 

tdown       =   2.10 

ttop        =  10.00 

Xini        =  20.00 

Xleft       = (Ywater - Yfoundation)*tup + Xini 

Xleft1      = (Ytop - Yfoundation)*tup + Xini 

Xright      = Xleft1 + ttop + (Ytop - Yfoundation)*tdown 

Xexit       = Xleft1 + ttop + (Ytop - Yexit)*tdown 

XEndPnt     = 200.00 

XbaseR      = 500.00 

 

X1  = 0          $ Y1  = 0 

X2  = XbaseR     $ Y2  = 0 

X3  = 0          $ Y3  = Yfoundation 

X4  = Xini       $ Y4  = Y3 

X5  = Xright     $ Y5  = Y3 

X6  = X2         $ Y6  = Y3 

X7  = Xleft      $ Y7  = Ywater 

X8  = Xleft1     $ Y8  = Ytop 

X9  = X8 + ttop  $ Y9  = Y8 

X10 = Xexit      $ Y10 = Yexit 

 

! KP's definition 

*DO,II,1,10 

  K,II,X%II%,Y%II% 

*ENDDO 

/PNUM,KP,1 

/REPLOT 

KPLOT 

 

! Generate areas (layers) 

A,1,2,6,5,4,3             ! Area 1 

A,5,10,9,8,7,4            ! Area 2 

 

! Meshing 

AESIZE,ALL,10             ! Element size = 3m 

  

! Mesh layers 

MAT,2 

AMESH,1                   ! Meshing Area 1 

 

AESIZE,ALL,6 

MAT,1 

AMESH,2                   ! Meshing Area 2 

 

ALLSEL,ALL 

EPLOT 

5.5 RESULTS  

 

In the following pages graphs with the obtained results are shown.  

 

5.5.1 Seepage through the dam  

 

 

These results may vary slightly because the free mesh

Flow vectors [(m^3/s)/m^2] 

Total flux (m^3/s)

X

Y

Z

.464E-15    

0

(x10**-8)       

-2.787

-2.322

-1.393

-.929

0

(x10**-7)       

Flux

0

14.966

44.898

59.864

74.83

89.796

104.762

119.728

134.694

149.659

Length [m]

 Slope stability analysis                                                       

Stability by Bishop’s method  

 

 

A safety factor of 1.8 for the worst sliding surface is obtained.  

It is important that the point of the center’s grid that determines the smaller safety 
factor is not located in the boundary of this mesh, because this would imply that 
points located outside this contour could still have lower safety factors. In this 
example, expanding the grid would not provide significantly smaller results of the 
safety factor, reason why the obtained safety factor will be considered acceptable.  

 

5.5.2 Stability by Janbu’s method  

 

 

 

In this plot, the sliding surface can be seen as a violet outline. This sliding surface 
provides a safety factor of 1.9.  

5.5.3 Total stresses and effective pressures (FEM)  

 

 

 

FEM: Vertical stresses N/m
2

MN

MX

X

Y

Z

-.376E+07   

-.334E+07   

-.293E+07   

-.251E+07   

-.209E+07   

-.167E+07   

-.125E+07   

-833634     

-415118     

3397        

FEM: Effective pressure contour N/m
2

MN

MX

X

Y

Z

-197011     

218754      

634519      

.105E+07    

.147E+07    

.188E+07    

.230E+07    

.271E+07    

.313E+07    

.354E+07    

5.5.4 FEM Safety Factor  

 

 

 

5.5.5 Values comparison 

 

Analysing the obtained results by the different methods, the following values are 
obtained 

 

Method  F  

Bishop  1.8  

Janbu  1.9  

FEM  1.9  

5.6 SUMMARY  

 

With this exercise the user has had the occasion to practice and to understand the 
way in which CivilFEM treats the following subjects:  

 Seepage:  

o Handling of the seepage elements. 

o Obtaining of the saturation line.  

o Calculation of seepage flows.  

 Slope stability analysis: 

o Interaction between the seepage analysis and the slope stability 
calculation. 

o Handling some of the classical methods of stability analysis (Bishop 
and Janbu). 

o Calculation of the sliding safety factor from the finite elements analysis. 

The studied subjects cover a wide range of possibilities, frequently used in civil 
engineering, particularly in the analysis of earth dams or general slopes. 

